30 Chapter: Cayley Digraphs of Groups

The chapter assumes familiarity with the notation and material in Chapter 30 of [Gallian].

To work the exercises in this chapter you will first need to know how to create lists in GAP. A list is a collection of elements. The elements are enclosed within square brackets and are separated by commas. For example, the following creates the list consisting of the numbers 1, 2, 5 and 8.

```
gap> listexample:=[1,2,5,8];
[ 1, 2, 5, 8 ]
```

You can append additional elements to the end of the list using the command `Add`. For example, to add the number 11 to our list type:

```
gap> Add(listexample,11);
gap> listexample;
[ 1, 2, 5, 8, 11 ]
```

We can refer to the ith element in a list by typing the name of the list followed by `[i]`. For example:

```
gap> listexample[4];
8
gap> listexample[3]*20;
100
```

We can now use GAP to test whether we have a Hamiltonian circuit for a particular group and set of generators. In addition, we can get GAP to list the elements in this circuit in the order that they are traversed in the circuit (given a particular starting element).

A Hamiltonian circuit of D_4 with generators $R_{90} = (1,2,3,4)$ and $H = (1,2)(3,4)$ is obtained by applying $2*(3*R_{90}, H)$. [Gallian, Chapter 30, Example 3] We begin with a list containing only the identity:

```
gap> d:=[()];
[ () ]
```

Append elements to the list using the rule $2*(3*R_{90}, H)$.

```
gap> Add(d,(1,2,3,4)*d[1]);
gap> d;
[ (), (1,2,3,4) ]
gap> Add(d,(1,2,3,4)*d[2]);
gap> d;
[ (), (1,2,3,4), (1,3)(2,4) ]
gap> Add(d,(1,2,3,4)*d[3]);
gap> d;
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2) ]
gap> Add(d,(1,2)(3,4)*d[4]);
gap> d;
```



```
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4) ]
gap> Add(d,(1,2,3,4)*d[5]);
gap> d;
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4), (1,4)(2,3) ]
gap> Add(d,(1,2,3,4)*d[6]);
gap> d;
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4), (1,4)(2,3), (1,3) ]
gap> Add(d,(1,2,3,4)*d[7]);
gap> d;
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4), (1,4)(2,3), (1,3), (1,2)(3,4) ]
gap> Add(d,(1,2)(3,4)*d[8]);
gap> d;
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4), (1,4)(2,3), (1,3), (1,2)(3,4),
  () ]
```

Thus starting with the identity of D_4 and using the rule $2 * (3 * R_{90}, H)$ (where $R_{90} = (1,2,3,4)$ and $H = (1,2)(3,4)$) we get the Hamiltonian circuit:

$$\{e, (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4), (1,4)(2,3), (1,3), (1,2)(3,4), e\}.$$

We could reduce the number of repetitive operations needed to produce this circuit by writing a short program in GAP. The following program performs the same operations as the above but also allows you to choose which element you want to be the first element in the circuit. The first line of the program says it will take as input a single element `n`. This is the element of D_4 which we want to be the first element in the circuit. The next line defines local variables that will be used in the program. The next line starts the Hamiltonian circuit list. The list is called `s` and the first element in the list is set equal to the element `n` (the element that was input into the function). The next 9 lines of the program are a for-do loop that is performed 8 times (the order of D_4). The `elif` command means "else if". The `fi;` command ends the if-then-else statement. Similarly the `od;` command ends the for-do loop. The next to last line tells GAP to output the created list `s`.

```
CircuitCheck:= function(n)
local s,i;
s:=[n];
for i in [1..8] do
  if i = 4 then
     Add(s,(1,2)(3,4)*s[i]);
  elif i = 8 then
     Add(s,(1,2)(3,4)*s[i]);
  else
    Add(s,(1,2,3,4)*s[i]);
  fi;
od;
return s;
end;
```

Now we can read this program into GAP.


```
gap> Read("CircuitCheck");
gap> CircuitCheck(());
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2), (2,4), (1,4)(2,3), (1,3), (1,2)(3,4),
  () ]
gap> CircuitCheck((1,2,3,4));
[ (1,2,3,4), (1,3)(2,4), (1,4,3,2), (), (1,2)(3,4), (2,4), (1,4)(2,3), (1,3),
  (1,2,3,4) ]
```

Thus we see the Hamiltonian circuit $2*(3*R_{90},H)$ starting with the identity is

$$\{e,(1,2,3,4),(1,3)(2,4),(1,4,3,2),(2,4),(1,4)(2,3),(1,3),(1,2)(3,4),e\}$$

and the circuit starting with $(1,2,3,4)$ is

$$\{(1,2,3,4),(1,3)(2,4),(1,4,3,2),e,(1,2)(3,4),(2,4),(1,4)(2,3),(1,3),(1,2,3,4)\}$$

.

Exercises

30.1 Let $D_4 =< r,f \mid r^4 = e = f^2, rf = fr^{-1} >$. Write a short GAP program to verify that $6*[3*(r,0),(f,0),3*(r,0),(e,1)]$ is a Hamiltonian circuit in $\text{Cay}(\{(r,0),(f,0),(e,1)\} : D_4 \oplus \mathbf{Z}_6)$. [Gallian, Chapter 30, Exercise 9]

30.2 Use your program in Exercise 30.1 to find Hamiltonian circuits for $\text{Cay}(\{(r,0),(f,0),(e,1)\} : D_4 \oplus \mathbf{Z}_6)$ starting with the elements $(r,0)$, $(f,0)$ and $(e,1)$.

30.3 Use GAP to find a Hamiltonian circuit in $\text{Cay}(\{(a,0),(b,0),(e,1)\} : Q_4 \oplus Z_2)$. [Gallian, Chapter 30, Exercise 2]

30.4 Use GAP to find a Hamiltonian circuit in $\text{Cay}(\{(a,0),(b,0),(e,1)\} : Q_4 \oplus Z_4)$. Find a Hamiltonian circuit in $\text{Cay}(\{(a,0),(b,0),(e,1)\} : Q_4 \oplus Z_m)$ when m is even. [Gallian, Chapter 30, Exercise 3]

30.5 Use GAP to find a Hamiltonian circuit in $\text{Cay}(\{(a,0),(b,0),(e,1)\} : Q_4 \oplus Z_3)$. [Gallian, Chapter 30, Exercise 23]

30.6 Use GAP to find a Hamiltonian circuit in $\text{Cay}(\{(a,0),(b,0),(e,1)\} : Q_4 \oplus Z_5)$. Find a Hamiltonian circuit in $\text{Cay}(\{(a,0),(b,0),(e,1)\} : Q_4 \oplus Z_m)$ when m is odd. [Gallian, Chapter 30, Exercise 24]

